A Logical Labyrinth

The puzzle is taken from *The Lady or the Tiger* (by Smullyan 1982, formulated as IP-model by Martin J. Chlond, Cath M.Toase). The relevant chapter, *Ladies or Tigers*, contains 12 puzzles of increasing difficulty. In each puzzle a prisoner is faced with a decision where he must open one of several doors. In the first few examples each room contains either a lady or a tiger and in the more difficult examples rooms may also be empty. The following puzzle is the most difficult.

If the prisoner opens a door to find a lady he will marry her and if he opens a door to find a tiger he will be eaten alive. We assume that the prisoner would prefer to be married than eaten alive. It is also assumed that the lady is in some way special to the prisoner and he would prefer to find and marry her rather than open a door into an empty room.

Each of the doors has a sign bearing a statement that may be either true or false. The puzzle involves nine rooms. The statements on the nine doors are:

- Door 1: The lady is in an odd-numbered room
- Door 2: This room is empty
- Door 3: Either sign 5 is right or sign 7 is wrong
- Door 4: Sign 1 is wrong
- Door 5: Either sign 2 or sign 4 is right
- Door 6: Sign 3 is wrong
- Door 7: The lady is not in room 1
- Door 8: This room contains a tiger and room 9 is empty
- Door 9: This room contains a tiger and sign 6 is wrong

In addition, the prisoner is informed that only one room contains a lady; each of the others either contains a tiger or is empty. The sign on the door of the room containing the lady is true, the signs on all the doors containing tigers are false, and the signs on the doors of empty rooms can be either true or false.

The puzzle as stated does not have a unique solution until the prisoner is told whether or not room eight is empty and this knowledge enables him to find a unique solution.

Define subscripts $i = 1,\ldots,9$ and $j = 1,\ldots,3$ where (1 lady, 2 tiger, 3 empty) and as above variables are

$$x_{i,j} = \begin{cases} 1 & \text{if door } i \text{ hides prize } j \\ 0 & \text{otherwise} \end{cases}$$

$$t_i = \begin{cases} 1 & \text{if statement on door } i \text{ is true} \\ 0 & \text{otherwise} \end{cases}$$

1 Formulate the model as an IP model.

2 Use CPLEX to solve the model.

3 What was the additional information the prisoner needed to know to find a unique solution?

Solution: A Logical Labyrinth

We will use the following arbitrary objective function

$$\max x_{1,1}$$

We now list the statements from the nine doors and state the relationship between the truth or falsity of each statement and the appropriate t_i variable. Linear constraints are developed in each case to enforce these relationships.

- Door 1: the lady is in an odd-numbered room.

  $$t_1 = 1 \Leftrightarrow x_{1,1} + x_{3,1} + x_{5,1} + x_{7,1} + x_{9,1} = 1$$

  This may be enforced by

  $$t_1 = x_{1,1} + x_{3,1} + x_{5,1} + x_{7,1} + x_{9,1}$$

- Door 2: This room is empty.

  $$t_2 = 1 \Leftrightarrow x_{2,3} = 1$$

  enforced by

  $$t_2 = x_{2,3}$$

- Door 3: Either sign 5 is right or sign 7 is wrong.

  $$t_3 = 1 \Leftrightarrow t_5 + (1 - t_7) \geq 1$$

  enforced by

  $$t_5 - t_7 - 2t_3 \leq -1$$
  $$t_5 - t_7 - t_3 \geq -1$$

- Door 4: Sign 1 is wrong.

  $$t_4 = 1 \Leftrightarrow t_1 = 0$$

  enforced by

  $$t_4 = 1 - t_1$$

- Door 5: Either sign 2 or sign 4 is right.

$$t_5 = 1 \Leftrightarrow t_2 + t_4 \geq 1$$

  enforced by

$$t_2 + t_4 - 2t_5 \leq 0$$
$$t_2 + t_4 - t_5 \geq 0$$

- Door 6: Sign 3 is wrong.

$$t_6 = 1 \Leftrightarrow t_3 = 0$$

$$t_6 = 1 - t_3$$

- Door 7: The lady is not in room 1.

$$t_7 = 1 \Leftrightarrow x_{1,1} = 0$$

  enforced by

$$t_7 = 1 - x_{1,1}$$

- Door 8: This room contains a tiger and room 9 is empty.

$$t_8 = 1 \Leftrightarrow x_{8,2} + x_{9,3} \geq 2$$

  enforced by

$$x_{8,2} + x_{9,3} - 2t_8 \leq 1$$
$$x_{8,2} + x_{9,3} - 2t_8 \geq 0$$

- Door 9: This room contains a tiger and sign 6 is wrong

$$t_9 = 1 \Leftrightarrow x_{9,2} + (1 - t_6) \geq 2$$

  enforced by

$$x_{9,2} - t_6 - 2t_9 \geq -1$$
$$x_{9,2} - t_6 - t_9 \leq 0$$

Further conditions of the puzzle are modeled as follows:

- Each door hides one prize

$$\sum_{j=1}^{3} x_{i,j} = 1 \text{ for } i = 1, \ldots, 9$$

- Only one room contains a lady.

$$\sum_{i=1}^{9} x_{i,1} = 1$$

- The sign on the lady's door is true.

$$t_i \geq x_{i,1} \text{ for } i = 1, \ldots, 9$$

- The sign on the tigers doors are false.

$$t_i \leq 1 - x_{i,2} \text{ for } i = 1, \ldots, 9$$

Experimentation with the model reveals that if the prisoner had been told that room eight was empty he could not have identified the location of the lady, as we have the following two solutions:

$x_{11} = 1$ $t_1 = 1$	$x_{22} = 1$	$x_{33} = 1$ $t_3 = 1$	$x_{43} = 1$	$x_{53} = 1$	$x_{63} = 1$	$x_{73} = 1$	$x_{83} = 1$	$x_{93} = 1$
$x_{13} = 1$ $t_1 = 1$	$x_{22} = 1$	$x_{33} = 1$	$x_{43} = 1$	$x_{53} = 1$	$x_{63} = 1$ $t_6 = 1$	$x_{71} = 1$ $t_7 = 1$	$x_{83} = 1$	$x_{93} = 1$

He must therefore have been informed that room eight was not empty. This additional feature requires the constraint

$$x_{8,3} = 0$$

and the revised model uniquely identifies the whereabouts of the lady as follows:

$x_{13} = 1$ $t_1 = 1$	$x_{22} = 1$	$x_{33} = 1$	$x_{43} = 1$	$x_{53} = 1$	$x_{63} = 1$ $t_6 = 1$	$x_{71} = 1$ $t_7 = 1$	$x_{82} = 1$	$x_{92} = 1$

```
maximize
  x11

subject to
  t1 - x11 - x31 - x51 - x71 - x91 = 0
  t2 - x23 = 0
  t5 - t7 - 2t3 <= -1
  t5 - t7 - t3 >= -1
  t4 + t1 = 1
  t2 + t4 - 2t5 <= 0
  t2 + t4 - t5 >= 0
  t6 + t3 = 1
  t7 + x11 = 1
  x82 + x93 - 2t8 <= 1
  x82 + x93 - 2t8 >= 0
  x92 - t6 - 2 t9 >= -1
  x92 - t6 - t9 <= 0

  x11 + x12 + x13 = 1
  x21 + x22 + x23 = 1
  x31 + x32 + x33 = 1
  x41 + x42 + x43 = 1
  x51 + x52 + x53 = 1
  x61 + x62 + x63 = 1
  x71 + x72 + x73 = 1
  x81 + x82 + x83 = 1
  x91 + x92 + x93 = 1

  x11 + x21 + x31 + x41 + x51 + x61 + x71 + x81 + x91 = 1

  t1 - x11 >= 0
  t2 - x21 >= 0
  t3 - x31 >= 0
  t4 - x41 >= 0
  t5 - x51 >= 0
  t6 - x61 >= 0
  t7 - x71 >= 0
```

```
  t8 - x81 >= 0
  t9 - x91 >= 0

  t1 + x12 <= 1
  t2 + x22 <= 1
  t3 + x32 <= 1
  t4 + x42 <= 1
  t5 + x52 <= 1
  t6 + x62 <= 1
  t7 + x72 <= 1
  t8 + x82 <= 1
  t9 + x92 <= 1

  x83 = 0

binary
  t1 t2 t3 t4 t5 t6 t7 t8 t9
  x11 x12 x13
  x21 x22 x23
  x31 x32 x33
  x41 x42 x43
  x51 x52 x53
  x61 x62 x63
  x71 x72 x73
  x81 x82 x83
  x91 x92 x93

end
```